

# Technische Informatik 1 – Übung 5: Eingabe/Ausgabe (Computerübung)

Olga Saukh  
10. & 11. November 2011





# Inhalt

- **Implementierung von Device-I/O mittels Polling und Interrupts**
- **Benötigte Grundlagen**
  - **Assembler**
  - **Zustandsdiagramm**
- **Aufgaben**
  - **Bit-Test**
  - **Polling**
  - **Interrupts (Zusatzaufgabe)**



# Aufgabe 1: Bit-Test: Aufgaben

- Zum Überprüfen des Status eines I/O- Devices muss oft ein einziges Bit geprüft werden. Implementieren Sie in **C und Assembler** dafür eine Funktion

```
int bittest(int x, int n)
```

bittest soll 1 zurückgeben, falls das Bit an der Position  $n$  in  $x$  gesetzt ist, andernfalls 0.

- Hinweis:

- **sllv** (shift left logical variable)

```
sllv $rd, $rs, $rt:    $rd = $rs << $rt
```

- **srlv** (shift right logical variable)

```
srlv $rd, $rs, $rt:    $rd = $rs >> $rt
```

# Aufgabe 1: Bit-Test: Lösung

```
int bittest(int x, int n) {  
    if (x & (1 << n)) {  
        return 1;  
    }  
    return 0;  
}
```

**bittest:**

```
li    $t0, 1  
sllv  $t0, $t0, $a1  
and   $v0, $a0, $t0  
bgtz  $v0, setone  
jr    $ra
```

**setone:**

```
li    $v0, 1  
jr    $ra
```



## Aufgabe 2: Polling

- **Polling** bezeichnet das regelmässige Abfragen eines Statusregisters durch ein Programm (programmierte Ein-/Ausgabe).
- In dieser Aufgabe verwenden wir Polling, um ein Zeichen vom Eingabe-Terminal zu lesen und es dann am Ausgabe-Terminal anzuzeigen (**Echo-Funktion**).



## Aufgabe 2: Polling: Aufgaben

- a) Zeichnen Sie ein **Zustandsdiagramm**, das die Echo-Funktion modelliert.
  
- b) Geben Sie ein **C Programmfragment** an, das ein Zeichen vom Receiver-Device liest. Verwenden Sie Polling, um auf die Verfügbarkeit eines Zeichens zu warten.
  
- c) Erstellen Sie ein **Assembler Programm**, das die Echo-Funktion implementiert. Verwenden Sie dazu das vorgegebene Template.

# Aufgabe 2: Polling: xspim -mapped\_io polling.s


```
.text
.globl main
main:
    addi    $sp, $sp, -4
    sw     $ra, 0($sp)
    lui    $t0, 0xffff

loop:
    #NEW CODE GOES HERE
    j     loop
```

The screenshot shows the xspim emulator interface. At the top, it displays the current PC (00400000), Status (3000ff10), and various registers (R0-R31). Below the registers, there are buttons for 'quit', 'load', 'reload', 'run', 'step', 'clear', 'set value', 'print', 'breakpoints', 'help', 'terminal', and 'mode'. The 'Text Segments' section shows the memory layout, including the main function code. The 'Data Segments' section shows the stack and kernel data. At the bottom, there is a copyright notice for SPIM Version 7.2.1 of August 28, 2005.

- Hinweis: “lw \$t1, x(\$t0)“ lädt das Wort an Adresse 0xFFFF0000 + x nach \$t1

# Aufgabe 2: Polling: Lösung (a)



## Aufgabe 2: Polling: Lösung (b)

```
volatile int *reg_read_ctrl = (int*)0xFFFF0000;  
volatile int *reg_read_data = (int*)0xFFFF0004;  
int character;
```

```
while(1) {  
    int status = *reg_read_ctrl;  
    if ((status & 0x1) == 0x1)  
        break;  
}  
character = *reg_read_data;
```

## Aufgabe 2: Polling: Lösung (c)

```
.text
```

```
.globl main
```

```
main:
```

```
addi $sp, $sp, -4
```

```
sw $ra, 0($sp)
```

```
lui $t0, 0xffff
```

```
loop:
```

```
poll_rx:
```

```
lw $t1, 0($t0)
```

```
andi $t1, $t1, 1
```

```
beqz $t1, poll_rx
```

```
lw $t2, 4($t0)
```

```
poll_tx:
```

```
lw $t3, 8($t0)
```

```
andi $t3, $t3, 1
```

```
beqz $t3, poll_tx
```

```
sw $t2, 12($t0)
```

```
j loop
```



## Aufgabe 3 – Interrupts (Zusatzaufgabe)

- Ein (Hardware-) **Interrupt** ist ein (externes) Signal, das den normalen Programmablauf unterbricht.
- Verwendung von Interrupts:
  - Vermeidung von Polling Loops
  - Time-/Event-Driven Multi-Tasking
- In dieser Aufgabe verwenden wir Interrupts, um die Echo-Funktion zu implementieren.



## Aufgabe 3 – Interrupts: Aufgaben


- Lesen Sie **Kapitel A.7** in Patterson/Hennessy: “Computer Organization and Design”.
- a) Schreiben Sie ein Assembler Programm, das Interrupts für das Eingabe-Terminal **aktiviert**.
- b) Implementieren Sie einen **Interrupt Handler**, der das gerade eingegebene Zeichen im Ausgabe-Terminal anzeigt.

## Aufgabe 3 – Interrupts: Hinweise

- `mtc0` (move to coprocessor 0)  
`mtc0 $rs(on processor), $rd (on coprocessor):`  
 $\$rd = \$rs$

Beispiel (clear cause reg): `mtc0 $0, $13`

- Coprocessor 0 Status Register (Reg. Nr. 12):



- Verwenden Sie `save_t0 ... save_t4`, um Registerinhalte während der Service-Routine zu sichern.

## Aufgabe 3 – Interrupts: Lösung (a)

**# set bit 1 of receiver control register**

```
li    $t4, 2
```

```
sw    $t4, 0($t0)
```

**# create bitmask for interrupt status**

```
li    $t4, 0x0801
```

**# write status to status reg of coprocessor**

```
mtc0  $t4, $12
```

# Aufgabe 3 – Interrupts: Lösung (b)

```
.ktext 0x80000180
interrupt:
    .set noat
    sw    $at, save_at
    .set at

    sw    $t3, save_t3
    sw    $t2, save_t2
    sw    $t1, save_t1
    sw    $t0, save_t0

    lui   $t0, 0xffff
    lw    $t1, 4($t0)

poll:
    lw    $t2, 8($t0)
    andi  $t2, $t2, 1
    beqz  $t2, poll
    sw    $t1, 12($t0)
```

```
done:
    mtc0  $0, $13
    lw    $t0, save_t0
    lw    $t1, save_t1
    lw    $t2, save_t2
    lw    $t3, save_t3
    .set noat
    lw    $at, save_at
    .set at
    eret

.kdata
save_at: .word 0
save_t0: .word 0
save_t1: .word 0
save_t2: .word 0
save_t3: .word 0
```